// Copyright (C) 2017 The Android Open Source Project
//
// Licensed under the Apache License, Version 2.0 (the "License");
// you may not use this file except in compliance with the License.
// You may obtain a copy of the License at
//
//      http://www.apache.org/licenses/LICENSE-2.0
//
// Unless required by applicable law or agreed to in writing, software
// distributed under the License is distributed on an "AS IS" BASIS,
// WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
// See the License for the specific language governing permissions and
// limitations under the License.

apply plugin: 'maven-publish'
apply from: "$gradle.ext.exoplayerSettingsDir/missing_aar_type_workaround.gradle"

afterEvaluate {
    publishing {
        repositories {
            maven {
                url = findProperty('mavenRepo') ?: "${buildDir}/repo"
            }
        }
        publications {
            release(MavenPublication) {
                from components.release
                artifact androidSourcesJar
                groupId = 'com.google.android.exoplayer'
                artifactId = releaseArtifactId
                version releaseVersion
                pom {
                    name = releaseArtifactId
                    description = releaseDescription
                    licenses {
                        license {
                            name = 'The Apache Software License, Version 2.0'
                            url = 'http://www.apache.org/licenses/LICENSE-2.0.txt'
                            distribution = 'repo'
                        }
                    }
                    developers {
                        developer {
                            name = 'The Android Open Source Project'
                        }
                    }
                    scm {
                        connection = 'scm:git:https://github.com/google/ExoPlayer.git'
                        url = 'https://github.com/google/ExoPlayer'
                    }
                    withXml {
                        addMissingAarTypeToXml(it)
                    }
                }
            }
        }
    }
}
tasks.withType(PublishToMavenRepository) { it.dependsOn lint, test }

task androidSourcesJar(type: Jar) {
    archiveClassifier.set('sources')
    from android.sourceSets.main.java.srcDirs
}
